Numerična matematika v programskem jeziku Julia

Martin Vuk

2024

Predgovor

Ta knjiga vsebuje gradiva za izvedbo laboratorijskih vaj pri predmetu Numerična matematika na Fakulteti za računalništvo in informatiko Univerze v Ljubljani. Kljub temu je primerna za vse, ki bi se želeli bolje spoznati z uporabo numeričnih metod in se naučiti uporabljati programski jezik Julia. Predpostavljamo, da je bralec vešč programiranja v kakem drugem programskem jeziku.

Knjige o numerični matematiki se pogosto posvečajo predvsem matematičnim vprašanjem. Vsebina te knjige pa poskuša nasloviti bolj praktične vidike numerične matematike. Tako so primeri, če je le mogoče, povezani s problemom praktične narave s področja fizke, matematičnega modeliranja in računalništva, pri katerih lahko za rešitev problema uporabimo numerične metode.

Bralcu svetujemo, da vso kodo napiše in preskusi sam. Še bolje je, če kodo razširi in spreminja. Bralca spodbujamo, da se čim bolj igra z napisano kodo. Koda, ki je navedena v tej knjigi, je minimalna različica kode, ki reši določen problem in še ustreza minimalnim standardom pisanja kvalitetne kode. Pogosto izpustimo preverjanje ali implementacijo robnih primerov. Včasih opustimo obravnavo pričakovanih napak. Prednost smo dali berljivosti pred kompletnostjo, da je bralcu lažje razumeti, kaj koda počne.

Vaje so zasnovane za samostojno delo z računalnikom. Vsaka vaja se začne z opisom naloge in jasnimi navodili, kaj na naj bo končni rezultat. Nato sledijo podrobnejša navodila, kako se naloge lotiti. Na koncu je rešitev z razlago posameznih korakov. Na ta način lahko študentje rešijo nalogo z različno mero samostojnosti. Na koncu je rešitev naloge, ki ji lahko študentje sledijo brez dodatnega dela. Rešitev vključuje matematične izpeljave, programsko kodo in rezultate, ki jih dobimo, če programsko kodo uporabimo.

Domače naloge so brez rešitev in naj bi jih študenti oziroma bralec rešili povsem samostojno. Nekatere stare izpitne naloge so rešene, večina pa nima rešitve. Odločitev, da niso vključene rešitve za vse izpitne naloge je namerna, saj bralec lahko verodostojno preveri svoje znanje le, če rešuje tudi naloge, za katere nima dostopa do rešitev.

Kljub temu, da je knjiga namenjena študentom, je zasnovana tako, da je primerna za vse, ki bi se radi naučili uporabljati in implementirati osnovne algoritme numerične matematike. Primeri programov so napisani v programskem jeziku Julia.

Na tem mestu bi se rad zahvalil Bojanu Orlu, Emilu Žagarju, Petru Kinku in Aljažu Zalarju, s katerimi sem sodeloval pri numeričnih predmetih na FRI. Veliko idej za naloge, ki so v tej knjigi, prihaja prav od njih. Prav tako bi se zahvalil članom *Laboratorija za matematične metode v računalništvu in informatiki* posebej Neži Mramor-Kosta in Damirju Franetiču, ki so tako ali drugače prispevali k nastanku te knjige. Moja draga žena Mojca Vilfan je opravila delo urednika, za kar sem ji izjemno hvaležen. Na koncu bi se zahvalil študentom, ki so obiskovali numerične predmete, ki sem jih učil in so me naučili marsikaj novega.

Kazalo

1 Uvod v programski jezik Julia

V knjigi bomo za implementacijo algoritmov in ilustracijo uporabe izbrali programski jezik julia. Zavoljo učinkovitega izvajanja, uporabe dinamičnih tipov, funkcij specializiranih glede na signaturo in dobre podporo za interaktivno uporabo, je julia zelo primerna za implementacijo numeričnih metod in ilustracijo njihove uporabe. V nadaljevanju sledijo kratka navodila, kako začeti z `julio` in si pripraviti delovno okolje v katerem bo pisanje kode steklo čim bolj gladko.

Cilji tega poglavja so
- da se naučimo uporabljati Julio v interaktivni ukazni zanki,
- da si pripravimo okolje za delo v programskem jeziku Julia,
- da ustvarimo prvi paket in
- da ustvarimo prvo poročilo v formatu PDF.

Tekom te vaje bomo pripravili svoj prvi paket v Juliji, ki bo vseboval parametrično enačbo Geronove lemniskate. Napisali bomo teste, ki bodo preverili pravilnost funkcij v paketu. Nato bomo napisali skripto, ki uporabi funkcijo iz našega paketa in nariše sliko Geronove lemniskate. Na koncu bomo pripravili lično poročilo v formatu PDF.

1.1 Namestitev in prvi koraki

Sledite navodilom in namestite programski jezik Julia. Nato lahko v terminalu poženete ukaz `julia`. Ukaz odpre interaktivno ukazno zanko (angl. *Read Eval Print Loop* ali s kratico REPL) in v terminalu se pojavi ukazni poziv `julia>`. Za ukazni poziv lahko napišemo posamezne ukaze, ki jih nato `julia` prevede, izvede in izpiše rezultate. Poskusimo najprej s preprostimi izrazi

```
julia> 1 + 1
2

julia> sin(pi)
0.0

julia> x = 1; 2x + x^2
3
```

1.1.1 Funkcije

Funkcije lahko definiramo na več načinov.

```
julia> f(x) = x^2 + sin(x)
f (generic function with 1 method)

julia> f(pi/2)
3.4674011002723395
```

Za funkcije, ki zahtevajo več kode, uporabimo `function`.

```
julia> function g(x, y)
         z = x + y
         return z^2
       end
g (generic function with 1 method)
```

1.1.2 Vektorji in matrike

Vektorje lahko vnesemo z oglatimi oklepaji `[]`:

```
julia> v = [1, 2, 3]
3-element Vector{Int64}:
 1
 2
 3

julia> v[1] # prvi indeks je 1
1

julia> v[2:end] # zadnja dva elementa
2-element Vector{Int64}:
 2
 3

julia> sin.(v) # funkcije lahko apliciramo na elementih (operator .)
3-element Vector{Float64}:
 0.8414709848078965
 0.9092974268256817
 0.1411200080598672
```

Matrike vnesemo tako, da elemente v vrstici ločimo s presledki, vrstice pa s podpičji.

```
julia> M = [1 2 3; 4 5 6]
2×3 Matrix{Int64}:
 1  2  3
 4  5  6

julia> M[1, :] # prva vrstica
3-element Vector{Int64}:
 1
 2
 3
```

1.1.3 Paketi

Opomba! *Različni načini ukazne zanke*

Julia ukazna zanka (REPL) pozna več načinov, ki so namenjeni različnim opravilom.

- Osnovni način s pozivom `julia>` je namenjen vnosu kode v Juliji.
- Paketni način s pozivom `pkg>` je namenjen upravljanju s paketi. V paketni način pridemo, če vnesemo znak `]`.
- Način za pomoč s pozivom `help?>` je namenjen pomoči. V način za pomoč pridemo z znakom `?`.
- Lupinski način s pozivom `shell>` je namenjen izvajanju ukazov v sistemski lupini. V lupinski način vstopimo z znakom `;`.

1.2 Priprava delovnega okolja

Vnašanje ukazov v interaktivni zanki je lahko uporabno namesto kalkulatorja. Vendar je za resnejše delo bolje kodo shraniti v datoteke. Med razvojem, se datoteke s kodo nenehno spreminjajo, zato je treba kodo v interaktivni zanki vseskozi posodabljati. Paket Revise.jl poskrbi za to, da se nalaganje zgodi avtomatično, ko se datoteke spremenijo. Zato najprej namestimo paket *Revise* in poskrbimo, da se zažene ob vsakem zagonu interaktivne zanke.

Odprite julia in sledite ukazom spodaj.

```
julia> # pritisnemo ], da pridemo v paketni način
(@v1.10) pkg> add Revise
(@v1.10) pkg> # pritisnemo Backspace, da pridemo iz paketnega načina
julia> startup = """
         try
           using Revise
         catch e
           @warn "Error initializing Revise" exception=(e, catch_backtrace())
         end
       """
...
julia> path = homedir() * "/.julia/config") # ustvarimo direktorij
julia> write(path * "/startup.jl", startup) # zapišemo startup.jl
```

Okolje za delo z *julio* je pripravljeno.

Opomba! *Urejevalniki in programska okolja za* julio

Za lažje delo z datotekami s kodo potrebujete dober urejevalnik golega besedila, ki je namenjen programiranju. Če še nimate priljubljenega urejevalnika, priporočam VS Code in razširitev za Julio.

1.3 Priprava projektne mape

Pripravimo mapo, v kateri bomo hranili programe. Datoteke bomo organizirali tako, da bo vsaka vaja paket v svoji mapi, korenska mapa pa bo služila kot delovno okolje.

Pripravimo najprej korensko mapo. Imenovali jo bomo `nummat-julia`, lahko si pa izberete tudi drugo ime.

```
$ mkdir nummat-julia
$ cd nummat-julia
$ julia
```

Nato v korenski mapi pripravimo okolje s paketi in dodamo nekaj paketov, ki jih bomo potrebovali pri delu v interaktivni zanki.

```
julia > # pritisnemo ], da pridemo v način paketov
(@v1.10) pkg> activate . # pripravimo virtualno okolje v korenski mapi
(nummat-julia) pkg> add Plot # paket za risanje grafov
```

Priporočljivo je uporabiti tudi program za vodenje različic Git. Z naslednjim ukazom v mapi `nummat-julia` ustvarimo repozitorij za `git` in registriramo novo ustvarjene datoteke.

```
$ git init .
$ git add .
$ git commit -m "Začetni vpis"
```

Priporočam pogosto beleženje sprememb z `git commit`. Pogoste potrditve (angl. commit) olajšajo pregledovanje sprememb in spodbujajo k razdelitvi dela na majhne zaključene probleme, ki so lažje obvladljivi.

Opomba! Na mojem računalniku pa koda dela!

Izjava „Na mojem računalniku pa koda dela!“ je postala sinonim za problem ponovljivosti rezultatov, ki jih generiramo z računalnikom. Eden od mnogih faktorjev, ki vplivajo na ponovljivost, je tudi dostop do zunanjih knjižnic/paketov, ki jih naša koda uporablja in jih ponavadi ne hranimo skupaj s kodo. V `julii` lahko pakete, ki jih potrebujemo, deklariramo v datoteki `Project.toml`. Vsak direktorij, ki vsebuje datoteko `Project.toml` definira bodisi delovno okolje ali pa paket in omogoča, da preprosto obnovimo vse zunanje pakete, od katerih je odvisna naša koda.

Za ponovljivost sistemskega okolja, glej docker, NixOS in GNU Guix.

1.4 Priprava paketa za vajo

Ob začetku vsake vaje si bomo v mapi, ki smo jo ustvarili pred tem (`nummmat-julia`) najprej ustvarili direktorij oziroma paket za *julio*, kjer bo shranjena koda za določeno vajo. S ponavljanjem postopka priprave paketa za vsako vajo posebej, se bomo naučili, kako hitro začeti s projektom. Obenem bomo optimizirali način dela (angl. workflow), da bo pri delu čim manj nepotrebnih motenj.

```
$ cd nummat-julia
$ julia
julia> # pritisnemo ], da pridemo v način pkg
(@v1.10) pkg> generate Vaja00 # 00 bomo kasneje nadomestili z zaporedno številko vaje
(@v1.10) pkg> activate .
(nummat-julia) pkg> develop Vaja00
(nummat-julia) pkg> # pritisnemo tipko za brisanje nazaj, da pridemo v navaden način
julia>
```

Zgornji ukazi ustvarijo direktorij `Vaja00` z osnovno strukturo paketa v Jiliji. Za bolj obsežen projekt, ki ga želite objaviti, lahko uporabite PkgTemplates ali PkgSkeleton.

```
julia> cd("Vaja00") # pritisnemo ;, da pridemo v način lupine
shell> tree .
.
├── Project.toml
└── src
    └── Vaja00.jl

1 directory, 2 files
```

Direktoriju dodamo še teste, skripte z demnostracijsko kodo in README dokument.

```
shell> mkdir test
shell> touch test/runtests.jl
shell> touch README.md
shell> mkdir scripts
shell> touch scripts/demo.jl
shell> tree .
.
├── Manifest.toml
├── Project.toml
├── scripts
│   └── demo.jl
├── src
│   └── Vaja00.jl
└── test
    └── runtests.jl
```

> *Opomba! V okolju Windows lupina ne deluje najbolje*
>
> Zgornji ukazi v lupini v okolju Microsoft Windows mogoče ne bodo delovali. V tem primeru lahko mape in datoteke ustvarite v *raziskovalcu*. Lahko pa si okolje za *julio* ustvarite v Linuxu v Windowsih (WSL).

1.4.1 Geronova lemniskata

Ko je direktorij s paketom `Vaja00` pripravljen, lahko začnemo s pisanjem kode. Za vajo bomo narisali Geronove lemniskato. Najprej definiramo koordinatne funkcije

$$x(t) = \frac{t^2 - 1}{t^2 + 1} \qquad y(t) = 2\frac{t(t^2 - 1)}{(t^2 + 1)^2}. \tag{1.1}$$

Definicije shranimo v datoteki `Vaja00/src/Vaja00.jl`.

```
module Vaja00

"""Izračunaj `x` kordinato Geronove lemniskate."""
lemniskata_x(t) = (t^2 - 1) / (t^2 + 1)
"""Izračunaj `y` kordinato Geronove lemniskate."""
lemniskata_y(t) = 2t * (t^2 - 1) / (t^2 + 1)^2

# izvozimo imena funkcij, da so dostopna brez predpone `Vaja00`
export lemniskata_x, lemniskata_y
end # module Vaja00
```

Program 1: Vsebina datoteke `Vaja00.jl`.

V interaktivni zanki lahko sedaj pokličemo novo definirani funkciji.

```
import Pkg; Pkg.activate(".")
using Vaja00
lemniskata_x(1.2)
```

Nadaljujemo, ko se prepričamo, da lahko kličemo funkcije iz paketa `Vaja00`.

Kodo, ki bo sledila, bomo sedaj pisali v scripto `Vaja00\doc\demo.jl`.

```
using Vaja00
#' Krivuljo narišemo tako, da koordinati tabeliramo za veliko število parametrov.
t = range(-5, 5, 300) # generiramo zaporedje 300 vrednosti na [-5, 5]
x = lemniskata_x.(t) # funkcijo apliciramo na elemente zaporedja
y = lemniskata_y.(t) # tako da imenu funkcije dodamo .
#' Za risanje grafov uporabimo paket `Plots`.
using Plots
plot(x, y, label=false, title="Geronova lemniskata")
```

Program 2: Vsebina datoteke `demo.jl`

Skripto poženemo z ukazom:

```
include("Vaja00/doc/demo.jl")
```

Rezultat je slika lemniskate.



Slika 1: Geronova lemniskata

Opomba! *Poganjanje ukaz za ukazom v VsCode*

Če uporabljate urejevalnik VsCode in razširitev za Julio, lahko ukaze iz skripte poganjate vrstico za vrstico kar iz urejevalnika. Če pritisnete kombinacijo tipk `Shift + Enter`, se bo izvedla vrstica v kateri je trenutno kazalka.

1.5 Testi

V prejšnjem razdelku smo definirali funkcije in napisali skripto, s katero smo omenjene funkcije uporabili. Naslednji korak je, da dodamo teste, s katerimi preiskusimo pravilnost napisane kode.

Opomba! Avtomatsko testiranje programov

Pomembno je, da pravilnost programov preverimo. Najlažje to naredimo „na roke“, tako da program poženemo in preverimo rezultat. Testiranja „na roke“ ima veliko pomankljivosti od tega, da zahteva čas, da je lahko nekonsistentno, in dovzetno za napake. Alternativa ročnemu testiranju programov so avtomatski testi. Avtomatski testi so preprosti programi, ki izvedejo testirani program in rezultate preverijo. Avtomoatski testi so pomemben del agilnega razvoja programske opreme in omogočajo avtomatizacijo procesov razvoja programske, ki se imenuje nenehna integracija.

Uporabili bomo paket Test, ki olajša pisanje testov. Vstopna točka za teste je datoteka `test\runtests.jl`.

Avtomatski test je preprost program, ki pokliče določeno funkcijo in preveri rezultat.Najbolj enostavno je rezultat kar primerjati z v naprej znanim rezultatom, za katerega smo prepričani, da je pravilen. Uporabili bomo makroje @test in @testset iz paketa `Test`.

V datoteko `test/runtests.jl` dodamo teste za obe koordinatni funkciji, ki smo ju definirali:

```
using Vaja00, Test

@testset "Koordinata x" begin
  @test lemniskata_x(1.0) ≈ 0.0
  @test lemniskata_x(2.0) ≈ 3 / 5
end

@testset "Koordinata y" begin
  @test lemniskata_y(1.0) ≈ 0.0
  @test lemniskata_y(2.0) ≈ 12 / 25
end
```

Program 3: Testi za paket `Vaja00`

Za primerjavo rezultatov smo uporabili operator ≈, ki je alias za funkcijo isapprox.

Opomba! Primerjava števil s plavajočo vejico

Pri računanju s števili s plavajočo vejico se izogibajmo primerjanju števil z operatorjem == bit po bit. Pri izračunih pride do zaokrožitvenih napak. Zato se različni načini izračuna za isto število praviloma razlikujejo na zadnjih decimalkah v zapisu s plavajočo vejico. Na primer izraz `asin(sin(pi/4)) - pi/4` ne vrne točne ničle ampak vrednost `-1.1102230246251565e-16`, ki pa je zelo majhno število. Za primerjavo dveh vrednosti `a` in `b` zato ponavadi uporabimo izraz

$$|a - b| < \varepsilon, \tag{1.2}$$

kjer je ε večji, kot pričakovana zaokrožitvena napaka. Funkcija `isapprox` je namenjena ravno zgornji primerjavi.

Preden lahko poženemo teste, moramo ustvariti testno okolje. Sledimo priporočilom za testiranje paketov. V mapi `Vaje00/test` ustvarimo novo okolje in dodamo paket `Test`.

```
(nummat-julia) pkg> activate Vaje00/test
(test) pkg> add Test
(test) pkg> activate .
```

Teste poženemo tako, da v `pkg` načinu poženemo ukaz `test Vaja00`

```
(nummat-julia) pkg> test Vaja00
```

1.6 Dokumentacija

Dokumentacija programske kode je sestavljena iz različnih besedil in drugih virov, npr. videov, ki so namenjeni uporabnikom in razvijalcem programa ali knjižnice. Dokumentacija lahko vključuje komentarje v kodi, navodila za namestitev in uporabo programa in druge vire v raznih formatih z razlagami ozadja, teorije in drugih zadev povezanih s projektom. Dobra dokumentacija lahko veliko pripomore k uspehu določenega programa. Sploh to velja za knjižnice.

Tudi, če kode ne bo uporabljal nihče drug in verjamite, slabo dokumentirane kode, nihče ne želi uporabljati, bodimo prijazni do nas samih v prihodnosti in pišimo dobro dokumentacijo.

Pisali bomo 3 vrste dokumentacije:
- dokumentacijo posameznih funkcij in tipov,
- navodila za uporabnika v datoteki `README.md`,
- poročilo v formatu PDF

> *Opomba! Zakaj format PDF*
>
> Izbira formata PDF je mogoče presenetljiva za pisanje dokumentacije programske kode. V praksi so precej bolj uporabne HTML strani. Dokumentacija v obliki HTML strani, ki se generira avtomatično v procesu nenehne integracije je postala *de facto* standard.
>
> V kontekstu popravljanja domačih nalog in poročil na vajah pa ima format PDF še vedno prednosti. Saj ga je lažje pregledovati in popravljati.

1.6.1 Dokumentacija funkcij in tipov

Funkcije in tipe v Julii dokumentiramo tako, da pred definicijo dodamo niz z opisom funkcije. Več o tem si lahko preberete v priročniku za Julio.

1.6.2 Generiranje PDF poročila

Za pisanje dokumentacijo bomo uporabili format Markdown, ki ga bomo dodali kot komentarje v kodi. Knjižnica Weave.jl poskrbi za generiranje PDF poročila.

Za generiranje PDF dokumentov je potrebno namestiti TeX/LaTeX. Priporočam namestitev TinyTeX ali TeX Live, ki pa vsebuje vso izvorno kodo in zasede več prostora na disku. Po namestitvi programa TinyTex moramo dodamo še nekaj `LaTeX` paketov, ki jih potrebuje paket Weave. V terminalu izvedemo naslednji ukaz

```
tlmgr install microtype upquote minted
```

Poročilo pripravimo v obliki demo skripte. Uporabili bom kar `Vaja00/doc/demo.jl`, ki smo jo ustvarili, da smo generirali sliko.

V datoteko dodamo besedilo v obliki komentarjev. Komentarje, ki se začnejo z `#'` paket `Weave` uporabi kot tekst v formatu Markdown, medtem ko se koda in navadni komentarji v poročilu izpišejo kot koda.

```
#read("Vaja00/doc/demo.jl")
```

Program 4: Vsebina `Vaje00/doc/demo.jl`, po tem, ko smo dodali komentarje s tekstom v formatu Markdown

Poročilo pripravimo z ukazom `Weave.weave`. Ustvarimo še eno skripto `Vaje00\doc\makedocs.jl`, v katero dodamo naslednje vrstice

```
using Weave
# Poročilo generiramo z ukazom `Weave.weave`
Weave.weave("Vaje00/doc/demo.jl",
  doctype="minted2pdf", out_path="Vaja00/pdf")
```

Program 5: Program za generiranje PDF dokumenta

Skripto poženemo v julii s

```
include("Vaja00/doc/makedocs.jl").
```

Poleg paketa Weave.jl je na voljo še nekaj alternativ:

- IJulia,
- Literate.jl ali
- Quadro.

1.6.3 Povezave

Nekaj zanimivih povezav povezanih s pisanjem dokumentacije:

- Pisanje dokumentacije v jeziku Julia.
- Documenter.jl je najbolj razširjen paket za pripravo dokumentacije v Julii.
- Diátaxis je sistematičen pristop k pisanju dokumentacije.
- Dokumentacija kot koda je ime za način dela, pri katerem z dokumentacijo ravnamo na enak način, kot ravnamo s kodo.

1.7 Zaključek

Za konec preverimo, če vse dela kot bi moralo.

Ustvarili smo direktorij `nummat-julia` in direktorij s prvim paketom `nummat-julia\Vaja00`. V terminalu poženemo ukaz `tree .` da preverimo, če smo ustvarili vse datoteke.

```
$ tree .
nummat-julia
├── Project.toml
├── Manifest.toml
├── README.md
├── Vaja00
| ├── Project.toml
| ├── Manifest.toml
| ├── doc
| |  ├ makedocs.jl
| |  └ demo.jl
| ├ src
| | └ Vaja00.jl
| ├ pdf
| | ├ demo.pdf
| | └ ...
| └── test
|   └ runtests.jl
```

1.8 Povezave

- [Priporočila za stil Julia](https://docs.julialang.org/en/v1/manual/style-guide/).

- [Naveti za delo z Julijo](https://docs.julialang.org/en/v1/manual/workflow-tips/).

1.9 Git

[Git](https://git-scm.com/) je sistem za vodenje različic, ki je postal de facto standard v razvoju programske opreme pa tudi drugod, kjer se dela s tekstovnimi datotekami. Predlagam, da si bralec naredi svoj Git repozitorij, kjer si uredi kodo in zapiske, ki jo bo napisal pri spremljanju te knjige. Git repozitorij lahko hranimo zgolj lokalno na lastnem računalniku. Če želimo svojo kodo deliti ali pa zgolj hraniti varnostno kopijo, ki je dostopna na internetu, lahko repozitorij repliciramo lastnem strežniku ali na enm od javnih spletnih skladišč za programsko kodo npr. [Github](https://github.com/) ali [Gitlab](https://gitlab.com/).

1.9.1 Povezave

Spodaj je nekaj povezav, ki bodo bralcu v pomoč pri uporabi programa [Git](https://git-scm.com/):

- vmesnik za git [Git Extensions](https://gitextensions.github.io/) za Windows,
- [način dela za Github (Github flow)](https://docs.github.com/en/get-started/using-github/github-flow),
- [način dela za Gitlab (Gitlab Flow)](https://docs.gitlab.com/ee/topics/gitlab_flow.html).

2 Računanje kvadratnega korena

Računalniški procesorji navadno implementirajo le osnovne številske operacije: seštevanje, množenje in deljenje. Za računanje drugih matematičnih funkcij mora nekdo napisati program. Večina programskih jezikov vsebuje implementacijo elementarnih funkcij v standardni knjižnici. Tako tudi `julia`. V tej vaji si bomo ogledali, kako implementirati korensko funkcijo.

Opomba! Implementacija elementarnih funkcij v julii

Lokacijo metod, ki računajo določeno funkcijo lahko dobite z ukazoma `methods` in `@match`. Tako bo ukaz `methods(sqrt)` izpisal implementacije kvadratnega korena za vse podatkovne tipe, ki jih julia podpira. Ukaz `@which(sqrt(2.0))` pa razkrije metodo, ki računa koren za vrednost `2.0`, to je za števila s plavajočo vejico.

2.1 Naloga

Napiši funkcijo `y = koren(x)`, ki bo izračunala približek za kvadratni koren števila `x`. Poskrbi, da bo rezultat pravilen na 10 decimalnih mest in da bo časovna zahtevnost neodvisna od argumenta `x`.

2.1.1 Podrobna navodila

- Zapiši enačbo, ki ji zadošča kvadratni koren.
- Uporabi newtonovo metodo in izpelji Heronovo rekurzivno formulo za račnanje kvadratnega korena.
- Kako je konvergenca odvisna od vrednosti `x`?
- Nariši graf potrebnega števila korakov v odvisnoti od argumenta `x`.
- Uporabi lastnosti zapisa s plavajočo vejico in izpelji formulo za približno vrednost korena, ki uporabi eksponent (funkcija exponent v Julii).
- Implementiraj funkcijo `koren(x)`, tako da je časovna zahtevnost neodvisna od argumenta `x`. Grafično preveri, da funkcija dosega zahtevano natančnost za poljubne vrednosti argumenta `x`.

2.2 Računajne kvadratnega korena s Heronovim obrazcem

Z računajnjem kvadratnega korena so se ukvarjali že pred 3500 leti v Babilonu. O tem si lahko več preberete v članku v reviji Presek. Moderna verzija metode računanja približka predstavlja rekurzivno zaporedje, ki konvergira k vrednosti kvadratnega korena danega števila x. Zaporedje približkov lahko izračunamo, tako da uporabimo rekurzivno formulo

$$a_{n+1} = \frac{1}{2} \cdot \left(a_n + \frac{x}{a_n} \right). \tag{2.1}$$

Če izberemo začetni približek, zgornja formula določa zaporedje, ki vedno konvergira bodisi k $\sqrt{x}$ ali $-\sqrt{x}$, odvisno od izbire začetnega približka. Poleg tega, da zaporedje hitro konvergira k limiti, je program, ki računa člene izjemno preprost. Poglejmo si za primer, kako izračunamo $\sqrt{2}$:

```
#| output: true
let
    x = 1.5
    for n = 1:5
        x = (x + 2 / x) / 2
```

```
        println(x)
    end
end
```

Vidimo, da se približki začnejo ponavljati že po 4. koraku. To pomeni, da se zaporedje ne bo več spreminjalo in smo dosegli najboljši približek, kot ga lahko predstavimo z 64 bitnimi števili s plavajočo vejico.

Napišimo zgornji algoritem še kot funkcijo:

```
"""
  y = koren_heron(x, x0, n)

Izračuna približek za koren števila `x` z `n` koraki Heronovega obrazca z začetnim
približkom `x0`.
"""
function koren_heron(x, x0, n)
  y = x0
  for i = 1:n
    y = (y + x / y) / 2
    @info "Približek na koraku $i je $y"
  end
  return y
end
```

Program 6: Funkcija, ki računa kvadrani koren s Heronovim obrazcem.

Preskusimo funkcijo na številu 3.

```
  x = koren_heron(3, 1.7, 5)
  println("koren 3 je $(x)!")
```

```
[ Info: Približek na koraku 1 je 1.7323529411764707
[ Info: Približek na koraku 2 je 1.7320508339159093
[ Info: Približek na koraku 3 je 1.7320508075688776
[ Info: Približek na koraku 4 je 1.7320508075688772
[ Info: Približek na koraku 5 je 1.7320508075688772
koren 3 je 1.7320508075688772!
```

Opomba! *Metoda navadne iteracije in tangentna metoda*

Metoda računanja kvadratnega korena s Heronovim obrazcem je poseben primer tangentne metode, ki je poseben primer metode fiksne točke. Obe metodi, si bomo podrobneje ogledali, v poglavju o nelinearnih enačbah.

2.2.1 Izbira začetnega približka

Funkcija `koren_heron(x, x0, n)` ni uporabna za splošno rabo, saj mora uporabnik poznati tako začetni približek, kot tudi število korakov, ki so potrebni, da dosežemo željeno natančnost. Da bi lahko funkcijo uporabljal kdor koli, bi morala funkcija sama izbrati začetni približek, kot tudi število korakov.

Kako bi učinkovito izbrali dober začetni približek? Dokazati je mogoče, da rekurzivno zaporedje konvergira ne glede na izbran začetni približek. Tako lahko uporabimo kar samo število x. Malce boljši približek dobimo s Taylorjevem razvojem korenske funkcije okrog števila 1

$$\sqrt{x} = 1 + \frac{1}{2}(x-1) + \ldots \approx \frac{1}{2} + \frac{x}{2}. \tag{2.2}$$

Število korakov lahko izberemo avtomatsko tako, da računamo nove približke, dokler relativna napaka ne pade pod v naprej predpisano mejo (v našem primeru bomo izbrali napako tako, da bomo dobili približno 10 pravilnih decimalnih mest). Program implementiramo kot novo metodo za funkcijo `koren`

```
"""
    y, st_iteracij = koren_babilonski(x, x0)

Izračunaj vrednost kvadratnega korena danega števila `x` z babilonskim obrazcem z
začetnim približkom `x0`. Funkcija vrne
vrednost približka za kvadratni koren in število iteracij (kolikokrat zaporedoma smo
uporabili babilonski obrazec, da smo dobili zahtevano natančnost).
"""
function koren_babilonski(x, x0)
    a = x0
    it = 0
    while abs(a^2 - x) > abs(x) * 0.5e-11
        a = (a + x / a) / 2
        it += 1
    end
    return a, it
end

y, it = koren_babilonski(10, 0.5 + 10 / 2)
println("Za izračun korena števila 10, potrebujemo $it korakov.")
y, it = koren_babilonski(1000, 0.5 + 1000 / 2)
println("Za izračun korena števila 1000, potrebujemo $it korakov.")
```

Opazimo, da za večje število, potrebujemo več korakov. Poglejmo si, kako se število korakov spreminja, v odvisnosti od števila `x`.

```
#| fig-cap: Število korakov v odvisnosti od argumenta
using Plots
plot(x -> koren_babilonski(x, 0.5 + x / 2)[2], 0.0001, 10000, xaxis=:log10,
minorticks=true, formatter=identity, label="število korakov")
```

Začetni približek $\frac{1}{2} + \frac{x}{2}$ dobro deluje za števila blizu 1, če isto formulo za začetni približek preskusimo za večja števila, dobimo večjo relativno napako. Oziroma potrebujemo več korakov zanke, da pridemo do enake natančnosti. Razlog je v tem, da je $\frac{1}{2} + \frac{x}{2}$ dober približek za majhna števila, če pa se od števila 1 oddaljimo, je približek slabši, bolj kot smo oddaljeni od 1:

```
#| fig-cap: Začetni približek v primerjavi z dejansko vrednostjo korena.
using Plots
plot(x -> 0.5 + x / 2, 0, 10, label="začetni približek")
plot!(x -> sqrt(x), 0, 10, label="korenska funkcija")
```

Da bi dobili boljši približek, si pomagamo s tem, kako so števila predstavljena v računalniku. Realna števila predstavimo s števili s [plavajočo vejico](https://sl.wikipedia.org/wiki/Plavajo%C4%8Da_vejica). Število je zapisano v obliki

$$x = m2^e \tag{2.3}$$

kjer je $0.5 \leq m < 1$ mantisa, e pa eksponent. Za 64 bitna števila s plavajočo vejico se za zapis mantise uporabi 53 bitov (52 bitov za decimalke, en bit pa za predznak), 11 bitov pa za eksponent (glej [IEE 754 standard](https://en.wikipedia.org/wiki/IEEE_754)).

Koren števila x lahko potem izračunamo kot

$$\sqrt{x} = \sqrt{m} 2^{\frac{e}{2}} \tag{2.4}$$

dober začetni približek dobimo tako, da $\sqrt{m}$ aproksimiramo razvojem v Taylorjevo vrsto okrog točke 1

$$\sqrt{m} \approx 1 + \frac{1}{2}(m-1) = \frac{1}{2} + \frac{m}{2} \tag{2.5}$$

Če eksponent delimo z 2 in zanemarimo ostanek $e = 2d + o$, lahko $\sqrt{2^e}$ približno zapišemo kot

$$\sqrt{2^e} \approx 2^d. \tag{2.6}$$

Celi del števila pri deljenju z 2 lahko dobimo z binarnim premikom v desno (right shift). Potenco števila 2^n, pa z binarnim premikom števila 1 v levo za n mest. Tako lahko zapišemo naslednjo funckijo za začetni približek:

```
"""
    zacetni_priblizek(x)

Izračunaj začetni približek za tangentno metodo za računanje kvadratnega korena števila
`x`.
"""
function zacetni_priblizek(x)
    d = exponent(x) >> 1 # desni premik oziroma deljenje z 2
    m = significand(x)
    if d < 0
        return (0.5 + 0.5 * m) / (1 << -d)
    end
    return (0.5 + 0.5 * m) * (1 << d)
end
```

Primrjajmo izboljšano verzijo začetnega približka s pravo korensko funkcijo:

```
#| fig-cap: Izboljšan začetni približek.
using Plots
plot(zacetni_priblizek, 0, 1000, label="začetni približek")
plot!(sqrt, 0, 1000, label="kvadratni koren")
```

Oglejmo si sedaj število korakov, če uporabimo izboljšani začetni približek.

```
#| fig-cap: Število korakov v odvisnosti od argumenta za izboljšan začetni približek.

using Plots
plot(x -> koren_babilonski(x, zacetni_priblizek(x))[2], 0.0001, 10000, xaxis=:log10,
minorticks=true, formatter=identity, label="število korakov")
```

Opazimo, da se število korakov ne spreminja več z naraščanjem argumenta, to pomeni, da bo časovna zahtevnost tako implemetirane korenske funkcije konstantna in neodvisna od izbire argumenta.

```
#| fig-cap: Relativna napaka na [0.5, 2].
using Plots
rel_napaka(x) = (koren_babilonski(x, 0.5 + x / 2, 4)^2 - x) / x
plot(rel_napaka, 0.5, 2)
```

Sedaj lahko sestavimo funkcijo za računanje korena, ki potrebuje le število in ima konstantno časovno zahtevnost

```
"""
    y = koren(x)
```

```
Izračunaj kvadratni koren danega števila `x` z babilonskim obrazcem.
"""
function koren(x)
    y = zacetni_priblizek(x)
    for i = 1:4
        y = (y + x / y) / 2
    end
    return y
end
```

Preverimo, da je relativna napaka neodvisna od izbranega števila, prav tako pa za izračun potrebujemo enako število operacij.

```
#| fig-cap: Relativna napaka korenske funkcije.
plot(x -> (koren(x)^2 - x) / x, 0.001, 1000.0, xaxis=:log, minorticks=true,
formatter=identity, label="relativna napaka")
```

2.3 Hitro računanje obratne vrednosti kvadratnega korena

Pri razvoju računalniških iger, ki poskušajo verno prikazati 3 dimenzionalni svet na zaslonu, se veliko uporablja normiranje vektorjev. Pri operaciji normiranja je potrebno komponente vektorja deliti s korenom vsote kvadratov komponent. Kot smo spoznali pri računanju kvadratnega korena z babilonskim obrazcem, je posebej problematično poiskati ustrezen začetni približek, ki je dovolj blizu pravi rešitvi. Tega problema so se zavedali tudi inžinirji igre Quake, ki so razvili posebej zvit, skoraj magičen način za dober začetni približek. Metoda uporabi posebno vrednost `0x5f3759df`, da pride do začetnega približka, nato pa še en korak [tangentne metode](ttps: Več o [računanju obratne vrednosti kvadratnega korena](https://en.wikipedia.org/wiki/Fast_inverse_square_root).

3 Minimalne ploskve

3.1 Naloga

Žično zanko s pravokotnim tlorisom potopimo v milnico, tako da se nanjo napne milna opna.

Radi bi poiskali obliko milne opne, razpete na žični zanki. Malo brskanja po fizikalnih knjigah in internetu hitro razkrije, da ploskve, ki tako nastanejo, sodijo med minimalne ploskve, ki so burile domišljijo mnogim matematikom in nematematikom. Minimalne ploskve so navdihovale tudi umetnike npr. znanega arhitekta Otto Frei, ki je sodeloval pri zasnovi Muenchenskega olimpijskega stadiona, kjer ima streha obliko minimalne ploskve.

Slika 2: Slika olimpijskega stadiona v Münchnu.

3.2 Matematično ozadje

Ploskev lahko predstavimo s funkcijo dveh spremenljivk $u(x, y)$, ki predstavlja višino ploskve nad točko (x, y). Naša naloga bo poiskati funkcijo $u(x, y)$ na tlorisu žične mreže.

Funkcija $u(x, y)$, ki opisuje milno opno, zadošča matematična enačbi, znani pod imenom Poissonova enačba

$$\Delta u(x, y) = \rho(x, y) \tag{3.1}$$

Funkcija $\rho(x, y)$ je sorazmerna tlačni razliki med zunanjo in notranjo površino milne opne. Tlačna razlika je lahko posledica višjega tlaka v notranjosti milnega mehurčka ali pa teže, v primeru opne,

napete na žični zanki. V primeru minimalnih ploskev pa tlačno razliko kar zanemarimo in dobimo Laplaceovo enačbo:

$$\Delta u(x,y) = 0. \tag{3.2}$$

Če predpostavimo, da je oblika na robu območja določena z obliko zanke, rešujemo robni problem za Laplaceovo enačbo. Predpostavimo, da je območje pravokotnik $[a,b] \times [c,d]$. Poleg Laplacove enačbe, veljajo za vrednosti funkcije $u(x,y)$ tudi *robni pogoji*:

$$\begin{aligned} u(x,c) &= f_s(x) \\ u(x,d) &= f_z(x) \\ u(a,y) &= f_l(y) \\ u(b,y) &= f_d(y) \end{aligned} \tag{3.3}$$

kjer so f_s, f_z, f_l in f_d dane funkcije. Rešitev robnega problema je tako odvisna od območja, kot tudi od robnih pogojev.

Za numerično rešitev Laplaceove enačbe za minimalno ploskev dobimo navdih pri arhitektu Frei Otto, ki je minimalne ploskve raziskoval tudi z elastičnimi tkaninami.

3.3 Diskretizacija in linearni sistem enačb

Problema se bomo lotili numerično, zato bomo vrednosti $u(x,y)$ poiskali le v končno mnogo točkah: problem bomo *diskretizirali*. Za diskretizacijo je najpreprosteje uporabiti enakomerno razporejeno pravokotno mrežo točk na pravokotniku. Točke na mreži imenujemo *vozlišča*. Zaradi enostavnosti se omejimo na mreže z enakim razmikom v obeh koordinatnih smereh. Interval $[a,b])$ razdelimo na $n+1$ delov, interval $[c,d]$ pa na $m+1$ delov in dobimo zaporedje koordinat

$$\begin{aligned} a = x_0, x_1, \dots x_{n+1} = b \\ c = y_0,\ y_1, \dots y_{m+1} = d \end{aligned} \tag{3.4}$$

ki definirajo pravokotno mrežo točk $\big(x_i, y_j\big)$. Namesto funkcije $u : [a,b] \times [c,d] \to \mathbb{R}$ tako iščemo le vrednosti

$$u_{i,j} = u\big(x_i, y_j\big), \quad i = 1, \dots n, \quad j = 1, \dots m \tag{3.5}$$

Iščemo torej enačbe, ki jim zadoščajo elementi matrike $u_{i,j}$. Laplaceovo enačbo lahko diskretiziramo z končnimi diferencami, lahko pa izpeljemo enačbe, če si ploskev predstavljamo kot elastično tkanino, ki je fina kvadratna mreža iz elastičnih nitk. Vsako vozlišče v mreži je povezano s 4 sosednjimi vozlišči. Vozlišče bo v ravnovesju, ko bo vsota vseh sil nanj enaka 0. Predpostavimo, da so vozlišča povezana z idealnimi vzmetmi in je sila sorazmerna z razliko. Če zapišemo enačbo za komponente sile v smeri z, dobimo za točko $\big(x_i, y_j, u_{ij}\big)$ enačbo

$$u_{i-1,j} + u_{i,j-1} - 4u_{i,j} + u_{i+1,j} + u_{i,j+1} = 0. \tag{3.6}$$

Za u_{ij} imamo tako sistem linearnih enačb. Ker pa so vrednotsi na robu določene z robnimi pogoji, moramo elemente u_{0j}, $u_{n+1,j}$, u_{i0} in u_{im+1} prestaviti na desno stran in jih upoštevati kot konstante.

3.4 Matrika sistema linearnih enačb

Sisteme linearnih enačb običajno zapišemo v matrični obliki

$$A\boldsymbol{x} = \boldsymbol{b}, \tag{3.7}$$

kjer je A kvadratna matrika, $\boldsymbol{x}$ in $\boldsymbol{b}$ pa vektorja. Spremenljivke $u_{i,j}$ razvrstimo po stolpcih v vektor.

!!! note „Razvrstitev po stolpih“

```
Eden od načinov, kako lahko elemente matrike razvrstimo v vektor, je, da
stolpce matrike enega za drugim postavimo v vektor. Indeks v vektorju $k$ lahko
izrazimo z indeksi $i,j$ v matriki s formulo
$$k = i+(n-1)j.$$
```

Za $n = m = 3$ dobimo 9×9 matriko

```
L = \begin{bmatrix} -4& 1& 0& 1& 0& 0& 0& 0& 0\\ 1& -4& 1& 0& 1& 0& 0&
0& 0\\ 0& 1& -4& 0& 0& 1& 0& 0& 0\\ 1& 0& 0& -4& 1& 0& 1& 0& 0\\ 0& 1& 0& 1& -4&
1& 0& 1& 0\\ 0& 0& 1& 0& 1& -4& 0& 0& 1\\ 0& 0& 0& 1& 0& 0& -4& 1& 0\\ 0& 0& 0&
0& 1& 0& 1& -4& 1\\ 0& 0& 0& 0& 0& 1& 0& 1& -4\\ \end{bmatrix},
```

ki je sestavljena iz 3×3 blokov

```
\begin{bmatrix}-4&1&0\cr 1&-4&1\cr 0&1&-4\end{bmatrix},\quad
\begin{bmatrix}1&0&0\cr 0&1&0\cr 0&0&1\end{bmatrix}.
```

desne strani pa so

```
\mathbf{b} = -[u_{01}+u_{10}, u_{20}, \ldots u_{n0}+u_{n+1,1},u_{02},
0,\ldots u_{n+1,2}, u_{03}, 0\ldots u_{n, m+1},u_{n,m+1}+u_{n+1,m}]^T.
```

3.5 Izpeljava s Kronekerjevim produktom

Množenje vektorja $\boldsymbol{x} = \mathrm{vec}(Z)$ z matriko L lahko prestavimo kot množenje z matriko

$$\mathrm{vec}(LZ + ZL) = L\ \mathrm{vec}(Z). \tag{3.8}$$

Ker velja $\mathrm{vec}(AXB) = A \otimes B \cdot \mathrm{vec}(X)$ je

$$L^{N,N} = L^{m,m} \otimes I^{n,n} + I^{m,m} \otimes L^{n,n} \tag{3.9}$$

3.6 Primer

```
robni_problem = RobniProblemPravokotnik( LaplaceovOperator{2}, ((0,
pi), (0, pi)), [sin, y->0, sin, y->0] ) Z, x, y = resi(robni_problem) surface(x,
y, Z) savefig("milnica.png")
```

3.7 Napolnitev matrike ob eliminaciji

Matrika Laplaceovega operatorja ima veliko ničelnih elementov. Takim matrikam pravimo razpršene ali redke matrike. Razpršenost matirke lahko izkoristimo za prihranek prostora in časa, kot smo že videli pri tridiagonalnih matrikah. Vendar se pri Gaussovi eliminaciji delež ničelnih elementov matrike pogosto zmanjša. Poglejmo kako se odreže matrika za Laplaceov operator.

```
using Plots L = matrika(100,100, LaplaceovOperator(2)) spy(sparse(L),
seriescolor = :blues)
```

Če izvedemo eliminacijo, se matrika deloma napolni z neničelnimi elementi:

```
import LinearAlgebra.lu LU = lu(L) spy!(sparse(LU.L), seriescolor =
:blues) spy!(sparse(LU.U), seriescolor = :blues)
```

3.8 Koda

```
@index Pages = ["03_minimalne_ploskve.md"]
@autodocs Modules = [NumMat] Pages = ["Laplace2D.jl"]
```

3.9 Iteracijske metode

```
@meta
CurrentModule = NumMat
DocTestSetup  = quote
    using NumMat
end
```

V nalogi o minimalnih ploskvah smo reševali linearen sistem enačb

```
u_{i,j-1}+u_{i-1,j}-4u_{ij}+u_{i+1,j}+u_{i,j+1}=0
```

za elemente matrike $U = \left[u_{ij}\right]$, ki predstavlja višinske vrednosti na minimalni ploskvi v vozliščih kvadratne mreže. Največ težav smo imeli z zapisom matrike sistema in desnih strani. Poleg tega je matrika sistema L razpršena (ima veliko ničel), ko izvedemo LU razcep ali Gaussovo eliminacijo, veliko teh ničelnih elementov postane neničelni in matrika se napolni. Pri razpršenih matrikah tako pogosto uporabimo iterativne metode za reševanje sistemov enačb, pri katerih matrika ostane razpršena in tako lahko prihranimo veliko na prostorski in časovni zahtevnosti.

!!! note „Ideja iteracijskih metod je preprosta“

```
Enačbe preuredimo tako, da ostane na eni strani le en element s koeficientom 1. Tako
dobimo iteracijsko formulo za zaporedje približkov $u_{ij}^{(k)}$. Limita rekurzivnega
zaporedja je ena od fiksnih točk rekurzivne enačbo, če zaporedje konvergira. Ker smo
rekurzivno enačbo izpeljali iz originalnih enačb, je njena fiksna točka ravno rešitev
originalnega sistema.
```

V primeru enačb za laplaceovo enačbo(minimalne ploskve), tako dobimo rekurzivne enačbe

```
u_{ij}^{(k+1)} = \frac{1}{4}\left(u_{i,j-1}^{(k)}+u_{i-1,j}^{(k)}+u_{i+1,j}^{(k)}
+u_{i,j+1}^{(k)}\right),
```

ki ustrezajo jacobijevi iteraciji

!!! tip „Pogoji konvergence“

```
Rekli boste, to je preveč enostavno, če enačbe le pruredimo in se potem rešitel kar
sama pojavi, če le dovolj dolgo računamo. Gotovo se nekje skriva kak hakelc. Res je!
Težave se pojavijo, če zaporedje približkov **ne konvergira dovolj hitro** ali pa sploh
ne. Jakobijeva, Gauss-Seidlova in SOR iteracija **ne konvergirajo vedno**, zagotovo pa
konvergirajo, če je matrika po vrsticah [diagonalno dominantna](https://sl.wikipedia.
org/wiki/Diagonalno_dominantna_matrika).
```

Konvergenco jacobijeve iteracije lahko izboljšamo, če namesto vrednosti na prejšnjem približku, uporabimo nove vrednosti, ki so bile že izračunani. Če računamo element u_{ij} po leksikografskem vrstnem redu, bodo elementi $u_{il}^{(k+1)}$ za $l < j$ in $u_{lj}^{(k+1)}$ za $l < i$ že na novo izračunani, ko računamo $u_{ij}^{(k+1)}$. Če jih upobimo v iteracijski formuli, dobimo gauss-seidlovo iteracijo

$$u_{i,j}^{(k+1)} = \frac{1}{4}\left(u_{i,j-1}^{(k+1)} + u_{i-1,j}^{(k)} + u_{i+1,j}^{(k)} + u_{i,j+1}^{(k)}\right) \tag{3.10}$$

Konvergenco še izboljšamo, če približek $u_{ij}^{(k+1)}$, ki ga dobimo z gauss-seidlovo metodo, malce zmešamo s približkom na prejšnjem koraku $u_{ij}^{(k)}$

$$
\begin{aligned}
u_{i,j}^{(\mathrm{GS})} &= \frac{1}{4}\left(u_{i,j-1}^{(k+1)} + u_{i-1,j}^{(k)} + u_{i+1,j}^{(k)} + u_{i,j+1}^{(k)}\right) \\
u_{i,j}^{(k+1)} &= \omega u_{i,j}^{(\mathrm{GS})} + (1-\omega)u_{i,j}^{(k)}
\end{aligned}
\tag{3.11}
$$

in dobimo metodo SOR. Parameter ω je lahko poljubno število $(0, 2]$ Pri $\omega = 1$ dobimo gauss-seidlovo iteracijo.

3.9.1 Primer

```
using Plots
U0 = zeros(20, 20)
x = LinRange(0, pi, 20)
U0[1,:] = sin.(x)
U0[end,:] = sin.(x)
surface(x, x, U0, title="Začetni približek za iteracijo")
savefig("zacetni_priblizek.png")
```

```
L = LaplaceovOperator(2)
U = copy(U0)
animation = Animation()
for i=1:200
    U = korak_sor(L, U)
    surface(x, x, U, title="Konvergenca Gauss-Seidlove iteracije")
    frame(animation)
end
mp4(animation, "konvergenca.mp4", fps = 10)
```

```
@raw html <video width="600" height="400" controls> <source src="../konvergenca.mp4"
type="video/mp4"> <source src="konvergenca.mp4" type="video/mp4"> </video>
```

Konvergenca Gauss-Seidlove iteracije

3.9.2 Konvergenca

Grafično predstavi konvergenco v odvisnoti od izbire ω.

```
using Plots
n = 50
U = zeros(n,n)
U[:,1] = sin.(LinRange(0, pi, n))
U[:, end] = U[:, 1]
L = LaplaceovOperator(2)
omega = LinRange(0.1, 1.95, 40)
it = [iteracija(x->korak_sor(L, x, om), U; tol=1e-3)[2] for om in omega]
plot(omega, it, title = "Konvergenca SOR v odvisnosti od omega")
savefig("sor_konvergenca.svg")
```

3.9.3 Metoda konjugiranih gradientov

Ker je laplaceova matrika diagonalno dominantna z -4 na diagonali je negativno definitna. Zato lahko uporabimo metodo konjugiranih gradientov. Algoritem konjugiranih gradientov potrebuje le množenje z laplaceovo matriko, ne pa tudi samih elementov. Zato lahko izkoristimo možnosti, ki jih ponuja programski jezik `julia`, da lahko za isto funkcijo napišemo različne metode za različne tipe argumentov.

Preprosto napišemo novo metodo za množenje `*`, ki sprejme argumente tipa `LaplaceovOperator{2}` in `Matrix`. Metoda konjugiranih gradientov še hitreje konvergira kot SOR.

```
@example
using NumMat
n = 50
U = zeros(n,n)
U[:,1] = sin.(LinRange(0, pi, n))
U[:, end] = U[:, 1]
L = LaplaceovOperator{2}()
b = desne_strani(L, U)
Z, it = conjgrad(L, b, zeros(n, n))
println("Število korakov: $it")
```